

1. General Descriptions:

The documentation herein describes a small CPU designed for the teaching purpose. This CPU is used in conjunction with the ELE306 Laboratory number 7. In the Lab example, this CPU is used to control a simple traffic light. Read the program listing in the Section 5 of this document to gain more insight to the CPU operations.

2. Instruction Set Architecture (ISA) Design

Here are the ISA components for this CPU:

1. 8-bit data bus
2. 8-bit address bus
3. Memory mapped I/O addresses
4. One Accumulator and no general purpose register
5. Five instructions have been implemented (room to expand to 8 or 16 instructions)
6. Only two addressing modes
7. Only “z” flag or condition code was implemented
8. Equipped with an Adder/Subtractor but only addition was implemented.
9. Built-in parallel I/O ports: one 4-bit parallel output port and one 4-bit parallel input port
10. This CPU and the built-in I/O ports used 20% of Xilinx's 4010XL FPGA CLB's


2.1. Instruction Set

The following is the instruction set (five instructions):

Assembly	Operations	Instruction format	Opcode	Z	cycles
ADD \$M	$Acc \leftarrow Acc + M$	OP(3) unused M	001	Y	5
STR \$M	$M \leftarrow Acc$	OP(3) unused M	010	N	4
CLA	$Acc \leftarrow 0$	OP(3) unused	011	N	1
JNZ \$M	$PC \leftarrow M$ if $Z=0$	OP(3) unused M	101	N	3
RST	$PC \leftarrow 0$	OP(3) unused	111	N	1

2.2. Data Path

Details of the following data path block diagram can be verified from the schematic.



*The Acc is an 8-bit data register.

*A multiplexer is used for the MAR input.

2.3. Control Signals

There are 12 control signals. However, you will see only 11 in the table below since the loading of Acc is coincided with the loading of the Z flag (therefore CLA does not affect the Z flag).

Inst.	state	Micro-operations	Control signals										
			acc1	cla	mmx	mal	mdol	mdil	intrs	pcinc	pcl	irl	mem_w
IF	S1	MAR ← PC; MEM_RD;			1	1							
	S2	MDR ← MEM; PC ← PC+1;						1		1			
	S2	IR ← MDR;										1	
ADD	S4	MAR ← PC; MEM_RD;			1	1							
	S5	MDR ← MEM; PC ← PC+1;						1		1			
	S6	MAR ← MDR; MEM_RD;				1							
	S7	MDR ← MEM;						1					
	S8	Acc ← Acc + MDR;	1										
STR	S4	MAR ← PC; MEM_RD;			1	1							
	S5	MDR ← MEM; PC ← PC+1;						1		1			
	S6	MAR ← MDR; MDR ← Acc;				1	1						
	S7	MEM_WR;											1
CLA	S4	Acc ← "00";		1									
JNZ	S4	MAR ← PC; MEM_RD;			1	1							
	S5	MDR ← MEM; PC ← PC+1;						1		1			
	S6	If (Z = '0') PC ← MDR;									1		
RST	S4	PC ← "00";							1				

- AccI: load enable for Acc and is also the load enable for the Z flip-flop.
- Cla: reset signal to Acc
- Mmx: selecting signal to the MAR's multiplexer. 0 for MDR and 1 for PC
- Mal: load enable signal for MAR
- Mdol: load enable for the MDR output register (to memory/IO)
- Mdil: load enable for the MDR input register (from memory/IO)
- Intrst: the internal reset signal for the PC (to be merged with the external RST)
- Pcinc: enable the increment of PC
- Pcl: load enable for PC
- Irl: load enable for IR
- Mem_wr: read/write control to memory/IO; 0: read and 1: write


NOTE1: IF stands for instruction fetch. It occurs before the instruction (see below)

NOTE2: The first micro-operation of each instruction always starts at S4, because the IF takes three states to complete (see below).

3. Control Unit Design

The control unit is design as a finite state machine. It mission is simply to implement the control signals specified in the above table.

3.1. Flowchart



NOTE1: For simplicity, I listed only ADD (S4), ADD (S5), etc., you should refer to the micro-operations-to-control signals table above for the detailed micro-operations and control signal assignments.

NOTE2: For JNZ, the decision on Z is made at S6. So is the action if taken.

3.2. Finite State Machine

From the above flowchart, we know that we need a FSM with only eight states (the longest route goes through ADD instruction). My approach is to design a simple FSM that goes through eight states (S1 through S8) as indicated in the flowchart. For instance, at S4, if the Opcode is either "011" or "111" then the FSM should go right back to S1. At S6, check if the Opcode is "101" and go back to S1 if so. At S7, go back to S1 if Opcode is "010". Finally, always go back to S1 after S8. Design the FSM based on this and ignore all the control signals for now.

Once the FSM structure is established, we will establish the logic that generates the control signals. For instance, IRL is '1' only when in S3 and therefore

```
IRL <= '1' when (Sreg0=S3) else
      '0';
```

Here, Sreg0 is the internal state variable as found in the VHDL code. Similarly, MMX is '1' in the following conditions:

```
S1
Opcode="001" (ADD) and S4
Opcode="010" (STR) and S4
Opcode="101" (JNZ) and S4
```

A similar VHDL assignment can be constructed:

```
MMX <=      '1' when (Sreg0=S1) else
            '1' when (IR(7 downto 5)="001" and Sreg0=S4) else
            '1' when (IR(7 downto 5)="010" and Sreg0=S4) else
            '1' when (IR(7 downto 5)="101" and Sreg0=S4) else
            '0';
```

IR(7 downto 5) is of course the Opcode field of the IR!

3.3. State Diagram and HDL Editors


While the state diagram editor is quite convenient to generate the FSM structure, it is quite awkward to type in all the logic assignments. My preference is to use the state diagram editor to establish the FSM structure and then use the HDL editor to finish the logic assignment. Of course, the macro is then generated from the HDL editor. (You may check the Control.asf for the draft and compare it to the final Control.vhd)

4. Main Memory and I/O

For the demonstration purpose, the CPU is explicitly connected to the main memory and the parallel I/O making it a complete embedded system. There are three components: ROM, RAM and I/O here. The ROM contains the program that runs the traffic lights. The RAM is basically unused in this demonstration but maybe useful as a demonstration and/or lab assignment purpose.

4.1. Memory Map

Since I/O is memory mapped, all three memory components are found in the memory map. There are 64 bytes in ROM, 32 bytes in RAM and one byte address for the I/O (see below).



4.2. Interface

Since the ROM only needs six address bits, special attention is given to connect it to the CPU. Note that on the schematic, the address bus going into ROM needs to be named MEM_ADR[5:0] to indicate the desired connections to the main MEM_ADR[7:0] bus. Similar arrangement is needed for the RAM (in this case only five address bits). The I/O actually uses all eight address bits to decode.

A multiplexer is used here to select the appropriate memory input to the CPU. This is commonly implemented with tri-state buffers. However, a multiplexer will serve the same purpose and is much easier to negotiate when simulating the circuits. Note: A buffer is added to tap into the memory address bus for just one bit MEM_ADR6.

4.3. Parallel I/O

The Parallel I/O has a 4-bit input and a 4-bit output. Therefore, the higher 4 bits from the data bus were not used here. You may add them if you wish. The input and output ports share one memory address such that when write to address \$FF you write to the output port and when read from address \$FF you read from the input port.

4.4. Miscellaneous

The OR gates on the main schematic are used to merge the external reset (from an outside pin) and the internal reset (generated by the control unit when RST instruction executes). This will control the reset signal on PC. Also, the OSC4 is used here to provide clock signals. These clock signals are routed to the pins so clock frequency can be changed from outside jumpers. (NOTE: OSC4 allows only three outputs be used at the same time and the 8MHz output must be used)

5. Demonstration

The pin assignments:

Reset:	Pin 3	;the external reset signal, active high
Clock:	Pin 4	;the system clock input
OSC:	Pin 15 → 15Hz	
	Pin 16 → 16KHz	
	Pin 17 → 8MHz	
Output:	Pin 5 → Green LED	
	Pin 6 → Yellow LED (actually amber)	
	Pin 7 → Red LED	
	Pin 8 → unused	
Input	Pin 35 → Fast/Slow control	
	Pin 36 → unused	
	Pin 37 → unused	
	Pin 38 → unused	

The program listing (ROM contents):


Addr. Machine Codes


```


00  60          CLA          ;Clear Acc and then add the operand is
01  20 22      ADD $GDAT     ;equivalent to move the operand to Acc
03  40 FF      STR $IOP
05  60          CLA
06  20 25      ADD $ONE
08  20 FF      ADD $IOP
0A  A0 0E      JNZ $FG       ;checking fast/slow input
0C  60          CLA
0D  60          CLA
0E  60          FG:  CLA
0F  20 23      ADD $YDAT
11  40 FF      STR $IOP
13  60          CLA
14  20 24      ADD $RDAT
16  40 FF      STR $IOP
18  60          CLA
19  20 25      ADD $ONE
1B  20 FF      ADD $IOP
1D  A0 21      JNZ $RG       ;checking fast/slow input
1F  60          CLA
20  60          CLA
21  E0          RG:  RST      ;jump back to addr. 00

22  FE          GDAT: DAT    1111 1110  ;Constant to turn on Green LED
23  FD          YDAT: DAT    1111 1101  ;Constant for Yellow LED
24  FB          RDAT: DAT    1111 1011  ;Constant for Red LED
25  01          ONE:  DAT    0000 0001  ;Constant 1
          IOP:  EQU    $FF


```







ELE405 Spring 2001	Project: Tiny
Dept. Elec. & Computer Eng.	Macro: ZERO
University of Rhode Island	Date: 2/7/01
Date Last Modified: 2/12/1	Author:



Control

//diagram ACTIONS

IR[7:0]

mdol

mal

extrst

irl

cla

zflag

pcl

accl

clk

intrst


mdil

mmx

pcinc

mem_wr

Sreg0



```

1:  --
2:  -- File: C:\WINDOWS\DESKTOP\COURSES\405\DESIGNS\TINY\Control.vhd
3:  -- created: 02/09/01 08:34:58
4:  -- from: 'C:\WINDOWS\DESKTOP\COURSES\405\DESIGNS\TINY\Control.asf'
5:  -- bv fsm2hdl - version: 2.0.1.53
6:  --
7:  library IEEE;
8:  use IEEE.std_logic_1164.all;
9:
10: use IEEE.std_logic_arith.all;
11: use IEEE.std_logic_unsigned.all;
12:
13: library SYNOPSYS;
14: use SYNOPSYS.attributes.all;
15:
16: entity control is
17:     port (clk: in STD_LOGIC;
18:           extrst: in STD_LOGIC;
19:           IR: in STD_LOGIC_VECTOR (7 downto 0);
20:           zflag: in STD_LOGIC;
21:           accl: out STD_LOGIC;
22:           cla: out STD_LOGIC;
23:           intrst: out STD_LOGIC;
24:           irl: out STD_LOGIC;
25:           mal: out STD_LOGIC;
26:           mdil: out STD_LOGIC;
27:           mdol: out STD_LOGIC;
28:           mem_wr: out STD_LOGIC;
29:           mmx: out STD_LOGIC;
30:           pcinc: out STD_LOGIC;
31:           pcl: out STD_LOGIC);
32: end;
33:
34: architecture control_arch of control is
35:
36:
37:  -- SYMBOLIC ENCODED state machine: Sreg0
38:  type Sreg0_type is (S1, S2, S3, S4, S5, S6, S7, S8);
39:  signal Sreg0: Sreg0_type;
40:
41: begin
42:  --concurrent signal assignments
43:
44:
45:  Sreg0_machine: process (clk, extrst)
46:
47:  begin
48:
49:  if extrst='1' then
50:      Sreg0 <= S1;
51:  elsif clk'event and clk = '1' then
52:      case Sreg0 is
53:          when S1 =>
54:              Sreg0 <= S2;
55:          when S2 =>
56:              Sreg0 <= S3;
57:          when S3 =>
58:              Sreg0 <= S4;
59:          when S4 =>
60:              if (IR(7 downto 5)="011" or IR(7 downto 5)="111") then
61:                  Sreg0 <= S1;
62:              else
63:                  Sreg0 <= S5;
64:              end if;
65:          when S5 =>
66:              Sreg0 <= S6;
67:          when S6 =>
68:              if (IR(7 downto 5)="101") then
69:                  Sreg0 <= S1;
70:              else
71:                  Sreg0 <= S7;
72:              end if;
73:          when S7 =>
74:              if (IR(7 downto 5)="010") then
75:                  Sreg0 <= S1;
76:              else
77:                  Sreg0 <= S8;
78:              end if;
79:          when S8 =>
80:              Sreg0 <= S1;
81:          when others =>
82:              null;
83:      end case;
84:  end if;
85: end process;
86:
87:  --Control Signal Assignments
88:
89:  accl <= '1' when (IR(7 downto 5)="001" and Sreg0=S8) else
90:      '0';
91:
92:  cla <= '1' when (IR(7 downto 0)="01100000" and Sreg0=S4) else
93:      '0';
94:
95:  mmx <= '1' when (Sreg0=S1) else
96:      '1' when (IR(7 downto 5)="001" and Sreg0=S4) else
97:      '1' when (IR(7 downto 5)="010" and Sreg0=S4) else
98:      '1' when (IR(7 downto 5)="101" and Sreg0=S4) else
99:      '0';

```

```

98:  mal <= '1' when (Sreg0=S1) else
99:      '1' when (IR(7 downto 5)="001" and Sreg0=S4) else
100:      '1' when (IR(7 downto 5)="001" and Sreg0=S6) else
101:      '1' when (IR(7 downto 5)="010" and Sreg0=S4) else
102:      '1' when (IR(7 downto 5)="010" and Sreg0=S6) else
103:      '1' when (IR(7 downto 5)="101" and Sreg0=S4) else
104:      '0';
105:
106:  mdol <= '1' when (IR(7 downto 5)="010" and Sreg0=S6) else
107:      '0';
108:
109:  mdil <= '1' when (Sreg0=S2) else
110:      '1' when (IR(7 downto 5)="001" and Sreg0=S5) else
111:      '1' when (IR(7 downto 5)="001" and Sreg0=S7) else
112:      '1' when (IR(7 downto 5)="010" and Sreg0=S5) else
113:      '1' when (IR(7 downto 5)="101" and Sreg0=S5) else
114:      '0';
115:
116:  intrst<='1' when (IR(7 downto 5)="111" and Sreg0=S4) else
117:      '0';
118:
119:  pcinc<='1' when (Sreg0=S2) else
120:      '1' when (IR(7 downto 5)="001" and Sreg0=S5) else
121:      '1' when (IR(7 downto 5)="010" and Sreg0=S5) else
122:      '1' when (IR(7 downto 5)="101" and Sreg0=S5) else
123:      '0';
124:
125:  pcl <= '1' when (IR(7 downto 5)="101" and Sreg0=S6 and zflag='0') else
126:      '0';
127:
128:  irl <= '1' when (Sreg0=S3) else
129:      '0';
130:
131:  mem_wr<='1' when (IR(7 downto 5)="010" and Sreg0=S7) else
132:      '0';
133:
134:  end control_arch;

```

